J. Fac. Fish. Anim. Husb.,
Hiroshima Univ. (1978), **17**: 1 ~ 15

Determinations of Viscometric Constants in Empirical Flow Equations of Heated Starch Solutions

Kiyoshi KUBOTA, Yoshihiko HOSOKAWA, Kanichi SUZUKI and Hideaki HOSAKA

Department of Food Chemistry and Technology, Faculty of Fisheries and Animal Husbandry, Hiroshima University, Fukuyama

Received February 25, 1978
(Figs. 1–9, Tables 1–4, Appendix)

INTRODUCTION

The flow equations of liquid foods are important for designing various apparatuses and for the control of various plants. The flow equation of Newtonian fluids such as water and clear fruit juices is simple. For the laminar flow of Newtonian fluids, the flow through a tube has been formulized in the Hagen-Poiseuille equation which has only a theoretical basis. However, the flow of non-Newtonian fluids such as starch solutions and fruit purees is very complex for theoretical analysis, and, therefore, these flow equations have been defined as many empirical equations.

Fundamental viscometric data of non-Newtonian fluids have been obtained with various tube viscometers or rotational viscometers.

The rotational viscometers are relatively easy to operate, and some data have been presented by Charm and Merrill[1], Charm[2], Harper[3], Charm[4], Harper and El Sahrigi[5], Saravacos[6] and Hyman[7]. Tube viscometers have been chosen for the simplicity of their construction and their reliability of results. Eolkin[8], Charm[2], Ram and Tamir[9], Saravacos[10], Scalzo *et al.*[11], Scheve *et al.*[12], Rao *et al.*[13] and Rao and Bourne[14] employed various tube viscometers to determine the flow behavior and the empirical flow equations of non-Newtonian fluids.

In this paper, we used the capillary tube viscometer operated under various pressures, and we studied the viscometric behavior of heated starch solutions and the determinations of viscometric constants (flow constants) in empirical flow equations, using the non-linear least square method.

EMPIRICAL FLOW EQUATIONS

The flow equation of Newtonian fluids is expressed in the following fundamental formula:

$$\gamma = (1/\mu)\, g_c\, \tau \qquad (1)$$

where, γ (1/sec) is the shear rate, τ (g_f/cm^2) is the shear stress, μ($g/cm{\cdot}sec^2$) is the viscosity of fluids and g_c ($g{\cdot}cm/g_f{\cdot}sec^2$) is the gravitational conversion factor. The shear stress expressed in (g_f/cm^2) can be converted from or to (dyn/cm^2) by g_c.

The simply generalized flow equation of many non-Newtonian fluids is expressed as follows[15),16)]

$$\gamma = (1/K)(g_c\tau - g_c\tau_y)^n \tag{2}$$

where, K ($g^n/cm^n.sec^{2n-1}$) is the fluid consistency index, n (–) is the flow behavior index and τ_y (g_f/cm^2) is the yield stress (see Fig. 1).

Another popular equation is expressed as follows[17),18)]



Fig. 1. Flow of a non-Newtonian fluid in a circular tube.

$$g_c\tau = K'\gamma^{n'} + g_c\tau_y' \tag{2'}$$

where, K' ($g/cm{\cdot}sec^{2-n'}$), n' (–) and τ_y' (g_f/cm^2) are the viscometric constants. The values of K', n' and τ_y' in Eq. (2′) can be obtained as the values of $K^{1/n}$, $1/n$ and τ_y which used the viscometric constants in Eq. (2).

From Eq. (2), it is noted that Eq. (1) of Newtonian fluids is given by n=1 and τ_y=0, the power-law flow equation of pseudoplastic ($n > 1$) or dilatant ($n < 1$) fluids is given by τ_y=0, and the flow equation of Bingham plastic fluids is given by n=1.

The relation between the shear stress and the pressure difference acting on the cylinder of flowing fluid of radius r (cm) and rength L (cm) may be expressed by this formula :

$$2\pi rL\cdot\tau = \pi r^2\cdot\Delta P \qquad \therefore\ \tau = r\Delta P/2L \tag{3}$$

where, ΔP(g_f/cm^2) is the pressure difference. In a circular tube, Eq.(3) becomes

$$\tau_w = r_w\Delta P/2L \tag{4}$$

where, r_w (cm) is the radius of tube and τ_w (g_f/cm^2) is the shear stress at the wall.

Combining Eqs.(3) and (4), we get:

$$r = (\tau/\tau_w)\,r_w, \quad dr = (r_w/\tau_w)\,d\tau \tag{5}$$

The volumetric flow rate of flowing fluid may be expressed by

$$Q = \int_0^{r_w} 2\pi r\cdot u\cdot dr = \pi\int_0^{r_w} r^2\cdot\gamma\cdot dr \tag{6}$$

where, Q(cm^3/sec) is the volumetric flow rate and u(cm/sec) the velocity of fluids at the radius r.

Combining Eqs.(5) and (6), and integrating, we obtain:
For general non-Newtonian fluids:

$$Q = (\pi r_w^3 / \tau_w^3) \int_{\tau_y}^{\tau_w} \tau^2 \cdot \gamma \cdot d\tau = (2\pi g_c^n r_w^3 (\tau_w - \tau_y)^{n+1} / K\tau_w^3) \times ((\tau_w - \tau_y)^2 / 2(n+3) + \tau_y (\tau_w - \tau_y)/(n+2) + \tau_y^2 / 2(n+1)) \quad (7)$$

For power-law fluids:

$$Q = \pi g_c^n r_w^3 \tau_w^n / (n+3) K \quad (8)$$

For Bingham fluids:

$$Q = (\pi g_c r_w^3 \tau_w / K)(1/4 - (\tau_y / \tau_w)/3 + (\tau_y / \tau_w)^4 / 12) \quad (9)$$

For Newtonian fluids:

$$Q = \pi g_c r_w^3 \tau_w / 4K \quad (10)$$

The viscometric constant K in Eq.(10), K and n in Eq.(8) and K and τ_y in Eq. (9) are able to be solved by a linear least square method from the experimental data of volumetric flow rate and pressure difference, but the viscometric constants K, n and τ_y in Eqs.(7) are not able to be solved.

Therefore, we calculated the viscometric constants by a non-linear least square method using the digital electric computer (The Computation Center of Hiroshima Univ., HITAC 8700-OS7). The subroutine program of a non-linear least square method employed herein has been described in detail elsewhere[19].

The following standard deviation σ is minimized.

$$\sigma = (\sum_{i=1}^{m} (Q_{ods} - Q_{cal})_i^2 W_i / m)^{1/2} \quad (11)$$

where, Q_{obs} and Q_{cal} are the observed and calculated volumetric flow rate of fluid, m is the number of the experimental points and W_i is the weighing coefficient.

Initial values of viscometric constants were calculated by the following equations

$$n = 1, \qquad \tau_y = 0.1 (\tau_w)_{min}, \qquad K = \pi g_c r_w^3 (\tau_w)_{av} / 4 Q_{av} \quad (12)$$

where, $(\tau_w)_{min}$ and $(\tau_w)_{av}$ are the minimum and average values in the observed values of τ_w respectively, and Q_{av} is the average values of Q.

The flow chart for the determinations of the viscometric constants in the flow equations is shown in Fig. 2. The practical program for Eq.(7) and the calculated results of the 5 wt % sweet potato starch solution at 30°C are shown in the Appendix.



Fig. 2. Flow chart for the determinations of the viscometric constants.

EXPERIMENTAL

1. Apparatus

The capillary tube viscometer used in our experiments is shown in Fig. 3. All the experiments were made with a grass capillary tube of 0.143 cm inside diameter and 26.41 cm length. The diameters of the capillary tubes were calculated from the weight of mercury required to fill it throughly. The two cylinderical chambers which were used as a sample feeder and a reservoir had a capacity of 200 cm^3 (scaled capacity) respectively, and were connected through the capillary tube. At the bottom of the two chambers, the fluids can be mixed by means of a magnetic mixer. This characterized tube viscometer is useful for the study of the flow property on various suspensions.

The pressure of each chamber can be reduced by using a vacuum pump. The pressure was measured with a mercury manometer.

The volumetric flow rates of the fluid were measured from the time required for the scaled capacity on the chambers.

All measurements were made at constant temperatures from 30 to 70°C by using a constant-temperature water bath.

2. Material

The starch solutions used in this studies were prepared from the powdered starches by heating in 3 and 5 wt% starch solutions under vigorous agitation for 30 min at 90°C. The powdered starches were prepared commercially in Japan (wheat, corn, potato and



Fig. 3. Experimental apparatus

(1) Capillary tube, (2) Cylindrical chamber,
(3) Magnetic mixer, (4) Vacuum pump,
(5) Mercury manometer, (6) Buffer container.

sweet potato starches; Wako Pure Chemical Ind. Ltd.).

RESULTS AND DISCUSSION

The relations of the volumetric flow rate Q(cm^3/sec) and the pressure difference ΔP(g$_f$/cm^2) were determined in the capillary tube viscometer. The average of five determinations was recorded for each fluid. An pressure difference was corrected by subtracting $m_p \rho_s \bar{u}/g_c$ usually called the Hagenbach correction, where ρ_s(g/cm^3) and $\bar{u}$(cm/sec) are the density and the mean velocity of fluids. The density of the fluids was measured by using a pycrometer. m_p(–) is the Hagenbach coefficient which accounts for the contraction entrance effect from the large diameter of the chamber into the capillary tube. We used a value of m_p=2.0 which was obtained from the experiments with glycerin solutions (Katayama Chemical Ind. Ltd.; special grade) of known viscosity. The flow behavior of glycerin solutions in Newtonian. The comparisons of the observed values vs. the approved values[20] of the viscometric constant K(g/cm·sec) in Eq.(10) are shown in Table 1, and the observed and claculated values of the volumetric flow rates Q(cm^3/sec) vs. the shear stress at the wall τ_w (g$_f$/cm^2) are shown in Fig. 4. The shear stress at the wall τ_w for a given pressure difference ΔP can be calculated by Eq.(4).

The data of the volumetric flow rate Q vs. the shear stress at the wall τ_w of the heated starch solutions at various temperatures are plotted in Figs. 5–8. The curves in Figs. 5–8 are the calculated results as shown latter.

By applying the experimental volumetric flow rate and the shear stress at the wall in Figs. 5–8, it is possible to calculate the viscometric constants n, τ_y and K in Eqs.(7), (8), (9) and (10) respectively, using a non-linear least square method. The calculated values of these viscometric constants for some samples are listed in Table 2, and the comparisons of the observed values vs. the calculated values of the volumetric flow rate for these viscometric constants are tabulated in Table 3. The agreements between the observed results for Eq.(7) and Eq.(8) are satisfactory for all samples as shown for some samples in Table 3.

The values of τ_y increase with the increase in concentration from 3 wt% to 5wt%. It is necessary to take this into account for higher concentrated solutions. However, in the cases of the 3 and 5 wt% solutions used in this experiments, the values of τ_y in Eq.

Table 1. Comparisons of observed values vs. approved values of viscometric constant of glycerin solutions

Sample: Glycerin solutions (t=30°C)

Glycerin solutions x_w (wt %)	Viscometric constant K (g/cm·sec) observed values	approved values[20]
99.0	5.57	5.64
89.2	1.07	1.04
74.3	0.218	0.205



Fig. 4. Volumetric flow rate Q vs. shear stress at the wall τ_w of glycerin solutions
observed values:
glycerin solutions at t=30°C
x_w(wt%) = 99.0 △ 89.2 ○ 74.3 ▽
calculated values:
for approved values[20] in Table 1 ———



Fig. 5. Volumetric flow rate Q vs. shear stress at the wall τ_w of heated wheat starch solutions
observed values:

x_w(wt%)	t(°C) = 30	50	70
3	▽	○	△
5	▼	●	▲

calculated values:
for K in Table 4 ———



Fig. 6. Volumetric flow rate Q vs. shear stress at the wall τ_w of heated corn starch solutions.



Fig. 7. Volumetric flow rate Q vs. shear stress at the wall τ_w of heated potato starch solutions.



Fig. 8. Volumetric flow rate Q vs. shear stress at the wall τ_w of heated sweet potato starch solutions.

Table 2. Calculated values of viscometric constants of heated starch solutions

Viscometric constants	Flow equations			
	Eq. (7)	Eq. (8)	Eq. (9)	Eq. (10)
Sample: Weat starch solution (x_w=5 wt %, t = 30°C)				
n (−)	1.233	1.262	(1.0)	(1.0)
τ_y (g_f/cm^2)	0.00988	(0.0)	0.0675	(0.0)
K ($g^n/cm^n \cdot sec^{2n-1}$)	1.209	1.492	0.240	0.271
Sample: Corn starch solution (x_w=5 wt%, t=30°C)				
n (−)	1.226	1.529	(1.0)	(1.0)
τ_y (g_f/cm^2)	0.0833	(0.0)	0.134	(0.0)
K ($g^n/cm^n \cdot sec^{2n-1}$)	2.033	17.46	0.434	0.586
Sample: Potato starch solution (x_w=5 wt%, t=30°C)				
n (−)	1.195	1.319	(1.0)	(1.0)
τ_y (g_f/cm^2)	0.0445	(0.0)	0.117	(0.0)
K ($g^n/cm^n \cdot sec^{2n-1}$)	4.954	12.61	1.224	1.430
Sample: Sweet potato starch solution (x_w=5 wt%, t=30°C)				
n (−)	1.229	1.369	(1.0)	(1.0)
τ_y (g_f/cm^2)	0.0448	(0.0)	0.112	(0.0)
K ($g^n/cm^n \cdot sec^{2n-1}$)	3.302	9.285	0.668	0.804

( ): fixed values.

(7) may be overlooked and the values of n in Eq.(8) are found to be between 1.2 and 1.4. The values of K which fixed n=1.3 in Eq.(8) for all samples are listed in Table 4, and the calculated values for these constants are illustrated by the solid lines in Figs. 5–8. The calculated results for the 5 wt% corn starch solutions are larger than the observed results in the region of low τ_w. The reasons are that the values of τ_y were relatively large, and the values of n were larger than 1.3. Similar results have been observed on 5 and 7 wt% Pearl corn starch pastes by Scheve *et al.*[12]

The values of logarithm of K which fixed n=1.3 in Eq.(8) are plotted in Fig. 9 against the reciprocal of the absolute temperature. Nearly straight lines are obtained. These lines can be represented by the following equation of the Arrhenius form as shown by Charm[1], Scalzo *et al.*[11] and Heldman[18].

$$K = A \exp(B/T) = A \exp(E/R_g T) \tag{13}$$

where, T (°K)=t (°C) + 273.2 is the absolute temperature and R_g=1.987 cal/g-mol·°K is the gas constant. A, B and E are the constants which are calculated from the slopes and intercepts of the straight lines in Fig. 9. Table 4 shows the values of A and E for all samples steadied. Similar results have been observed on tomato concentrates by Harper and Sahrigi[5], on fruit juices and purees by Saravacos[6] and on egg and egg products by Salzo *et al.*[11]

This paper was presented on the occasion of the 25th Annual Meeting of the Japanese Society of Food Science and Technology, May 30, 1978, Tokyo.

Table 3. Comparisons of observed values vs. calculated values of volumetric flow rate of heated starch solutions

Shear stress at the wall τ_w(g_f/cm^2)	Volumetric flow rates Q_{obs} (cm^3/sec)	Q_{cal} (cm^3/sec) Eq.(7)	Eq.(8)	Eq.(9)	Eq.(10)
Sample: Wheat starch solution (x_W=5 wt%, t=30°C)					
0.102	0.374	0.446	0.485	0.177	0.852
0.263	1.735	1.588	1.599	1.630	2.190
0.540	3.884	3.975	3.957	4.222	4.492
0.816	6.667	6.673	6.655	6.803	6.781
1.091	9.615	9.603	9.610	9.387	9.074
Sample: Corn starch solution (x_W=5 wt%, t=30°C)					
0.176	0.258	0.192	0.305	0.090	0.676
0.260	0.414	0.496	0.555	0.456	0.999
0.376	0.998	0.990	0.977	1.037	1.446
0.656	2.401	2.376	2.285	2.477	2.520
1.001	4.320	4.330	4.364	4.265	3.848
Sample: Potato starch solution (x_W=5 wt%, t=30°C)					
0.260	0.273	0.246	0.252	0.197	0.410
0.432	0.506	0.513	0.491	0.509	0.681
0.783	1.065	1.131	1.076	1.153	1.233
1.184	1.830	1.913	1.854	1.889	1.864
1.498	2.548	2.569	2.530	2.467	2.359
Sample: Sweet potato starch solution (x_W=5 wt%, t=30°C)					
0.132	0.166	0.128	0.176	0.019	0.370
0.249	0.407	0.409	0.421	0.350	0.698
0.548	1.246	1.301	1.237	1.347	1.536
0.827	2.174	2.262	2.170	2.284	2.315
1.239	3.774	3.838	3.777	3.674	3.471

Table 4. Calculated values of viscometric constats of heated starch solutions

Sample solutions		Temperatures	Constant in Eq.(8)*	Constants in Eq.(13)*	
Starches	x_W(wt%)	t (°C)	K ($g^n/cm^n \cdot sec^{2n-1}$)	A($g^n/cm^n \cdot sec^{2n-1}$)	E(cal/g-mol)
Wheat	3	30, 50, 70	0.162, 0.137, 0.0941	1.63×10^{-3}	2.80×10^3
Wheat	5	30, 50, 70	1.914, 1.364, 0.819	1.39×10^{-3}	4.38×10^3
Corn	3	30, 50, 70	0.339, 0.225, 0.125	6.76×10^{-5}	5.16×10^3
Corn	5	30, 50, 70	3.952, 1.744, 1.033	3.54×10^{-5}	6.98×10^3
Potato	3	30, 50, 70	1.597, 0.787, 0.481	4.96×10^{-5}	6.24×10^3
Potato	5	30, 50, 70	11.086, 4.491, 2.936	1.04×10^{-4}	6.94×10^3
Sweet potato	3	30, 50, 70	1.050, 0.509, 0.374	1.28×10^{-4}	5.40×10^3
Sweet potato	5	30, 50, 70	5.856, 3.279, 2.049	6.80×10^{-4}	5.45×10^3

* : at n=1.3 fixed



Fig. 9. Effects of temperature on viscometric constant K in Table 4 of heated starch solutions.

observed values:

x_W(wt%)	wheat	corn	potato	sweet potato
3	□	△	○	▽
5	■	▲	●	▼

calculated values:
for A and E in Table 4

SUMMARY

The flow equations of liquid foods are important bases to design various apparatuses and to control various plants. The viscometric behavior of heated starch solutions was measured at 30, 50 and 70°C. The capillary tube viscometer was used in these studies. Heated starch solutions of 3 and 5 wt% wheat, corn, potato and sweet potato starches were used as samples.

An empirical flow equation $\gamma = (1/K)\ (g_c \tau - g_c \tau_y)^n$ was assumed, and the viscometric constants n, τ_y and K were calculated using the non-linear least square method. The yield stress τ_y (g_f/cm^2) could be overlooked for the heated starch solutions studied, and a power-law flow equation was applied.

The values of flow behavior index n(−) in the power-law flow equation were about 1.2 to 1.4 for all samples studied. Values of $K(g^n/cm^n \cdot sec^{2n-1})$ which fixed n=1.3 were obtained, and they were expressed by an equation of the Arrhenius from $K=A\ \exp(E/R_g T)$. The values of E were 3 to 7 kcal/g-mol for studied samples.

NOTATIONS

A, B and E	:	constants in Eq. (13)
g_c	:	gravitational conversion factor ($g \cdot cm/g_f \cdot sec^2$)
K	:	fluid consistency index ($g^n/cm^n \cdot sec^{2n-1}$)
L	:	length of capillary tube (cm)
m	:	number of experimental points (−)
m_p	:	Hagenbach coefficient (−)
n	:	flow behavior index (−)
ΔP	:	pressure difference (g_f/cm^2)
Q	:	volumetric flow rate of fluids (cm^3/sec)
R_g	:	gas constant (cal/g-mol·°K)
r	:	radius of capillary tube (cm)
T and t	:	temperature (°K) and (°C)
u and $\bar{u}$	:	velocity and average velocity of fluids (cm/sec)
x_w	:	weight percent of glycerin or starches (wt%)
γ	:	shear rate (1/sec)
μ	:	viscosity of fluids (g/cm·sec)
ρ_s	:	density of fluids (g/cm^3)
τ and τ_y	:	shear stress and yield stress (g_f/cm^2)

Subscripts:

w	:	at wall

obs, cal, min and av : observed, calculated, minimum and average values

REFERENCES

1) CHARM, S.E. and MERRILL, E.W.: *Food Res.*, **24**, 319–331 (1959).
2) CHARM, S.E.: *Food Res.*, **25**, 351–362 (1960).
3) HARPER, J.C.: *Food Tech.*, **14**, 557–561 (1960).
4) CHARM, S.E.: *J. Food Sci.*, **28**, 107-113 (1963).
5) HARPER, J.C. and EL SAHRIGI, A.F.: *ibid*, **30**, 470–476 (1965).

6) SARAVACOS, G.D.: *ibid,* **35**, 122–125 (1970).
7) HYMAN, W.A.: *Ind. Eng. Chem. Fundamentals,* **15**(3), 215–218 (1976).
8) EOLKIN, D.: *Food Tech.,* **11**, 253–257 (1957).
9) RAM, A. and TAMIR, A.: *Ind. Eng. Chem.,* **56**(2), 47–53 (1964).
10) SARAVACOS, G.D.: *Food Tech.,* **22**, 1585–1588 (1968).
11) SCALZO, A.M., DICKERSON, R.W., PEELER Jr., J.T. and READ Jr., R.B.: *Food Tech.,* **24**, 1301–1307 (1970).
12) SCHEVE, J.L., ABRAHAM, W.H. and LANCASTER, E.B.: *Ind. Eng. Chem. Fundamentals,* **13** (2), 150–154 (1974).
13) RAO, M.A., OTOYA PALOMINO, L.N. and BERNHARDT, L.W.: *J. Food Sci.,* **39**, 160–161 (1974).
14) RAO, M.A. and BOURNE, M.C.: *ibid,* **42**, 261–264 (1977).
15) ONOGI, S.: Reoroji Youron, pp. 9–12. Maki Shoten, Tokyo (1957).
16) HOSAKA, H., KUBOTA, K. and SUZUKI, K.: Shokuhin Kogaku, pp.1–17. Kyoritsu Shuppan, Tokyo (1975).
17) CHARM, S.E.: The Fundamentals of Food Engineering, pp.47–49. AVI, Westport, Connecticut (1963).
18) HELDMAN, D.R.: Food Process Engineering, pp.27–36. AVI, Westport, Connecticut (1975).
19) KUBOTA, K., SUZUKI, K. and HOSAKA, H.: *Shokuhin Kogyo,* **20** (16), 60–73 (1977).
20) LANGE, N.A.: Handbook of Chemistry, 10th Ed., p.1681. McGraw-Hill, New York (1967).

APPENDIX

```
C       MAIN PROGRAM
C       ESTIMATION OF NONLINEAR FROW PARAMETERS IN EQUATION
C       OF TUBE VISCOMETER USING NONLINEAR SUBROUTINE HISENS
C       N,L,K=,  NUMBER OF EXPERIMENTAL POINT, PARAMETER, ITERATION
C       RW,AL,V=,  RADIUS, LENGTH OF TUBE, VOLUME
C       DH(I),TS(I)=,  OBSERVED VALUES OF MANOMETER, TIME
C       DP(I),Q(1,I)=,  PRESSURE DROP, VOLUMETRIC FLOW RATE
C       U(I)=,  MEAN VELOCITY OF FLOW
C       TW(1,I)=,  SHEAR STRESS AT WALL
C       A(J,IK),SD(IK)=,  FROW PARAMETERS, STANDARD DEVIATION
C       QC(1,I,IK)=,  CALCULATED VALUES OF FLOW RATE
C       GC(I,IK)=, CALCULATED VALUES OF SHEAR RATE AT WALL
C       VISC(I,IK)=, CALCULATED VALUES OF APPARENT VISCOSITY
        DIMENSION DH(50),TS(50),DP(50),Q(1,50),TW(1,50),A(3,26),QC(1,50,
       126),QCS(1,50,4),GC(50,26),HM(26),HS(26),W(1,50),DD(3,4),SD(26)
       2,U(50),TWL(50),GCL(50,26),VISC(50,26)
        COMMON RW,GK
        DATA HAA/0.0005/,HM(1),HS(1)/2*1.0/,HMM,HSS/2*0.5/,HST/0.0001/,
       1EPS/1.0E-50/,MYT,MXT/2*1/,NT/50/,LT/3/,L1/4/,KT/25/,K1/26/
        EXTERNAL SUB
  888   READ(5,10) MY,MX,N,L
        IF(N) 999,999,777
  777   READ(5,20) RW,AL,V,ROUHG,ROUS
        AMP=2.0
        GK=980.0
        READ(5,30) (DH(I),I=1,N)
        READ(5,30) (TS(I),I=1,N)
        WRITE(6,50) MY,MX,N,L
        WRITE(6,51) RW,AL,V,ROUHG,ROUS
        WRITE(6,52) (DH(I),I=1,N)
        WRITE(6,53) (TS(I),I=1,N)
        TWAV=0.0
        QAV=0.0
        DO 100 I=1,N
        Q(1,I)=V/TS(I)
        U(I)=Q(1,I)/(3.1416*RW**2)
        DP(I)=DH(I)*ROUHG-AMP*ROUS*U(I)**2/GK
        TW(1,I)=RW*DP(I)/(2.0*AL)
        TWAV=TWAV+TW(1,I)
        QAV=QAV+Q(1,I)
        TWL(I)=ALOG10(TW(1,I))
```

```
100 W(1,I)=1.0
    TWAV=TWAV/N
    QAV=QAV/N
    WRITE(6,60) (DP(I),Q(1,I),U(I),TW(1,I),TWL(I),I=1,N)
    WRITE(6,65)
    A(1,1)=1.0
    A(2,1)=TW(1,1)/10.0
    A(3,1)=3.1416*GK*RW**3*TWAV/(4.0*QAV)
    WRITE(6,70) (A(J,1),J=1,L)
    HM(1)=1.0
    HS(1)=1.0
    CALL HISENS(Q,TW,A,HAA,HM,HMM,HS,HSS,HST,W,QC,QCS,DD,SD,
   1MY,MYT,MX,MXT,N,NT,L,LT,K,KT,L1,K1,EPS,SUB,ILL)
    IF(ILL) 200,300,200
200 WRITE(6,80) ILL
    GO TO 888
300 DO 400 I=1,N
    DO 400 IK=1,K1
    GC(I,IK)=GK*(TW(1,I)-A(2,IK))
    IF(GC(I,IK)) 500,510,510
500 GC(I,IK)=1.0E-5
510 GC(I,IK)=GC(I,IK)**A(1,IK)/A(3,IK)
    GCL(I,IK)=ALOG10(GC(I,IK))
400 VISC(I,IK)=GK*TW(1,I)/GC(I,IK)
    WRITE(6,90)
    WRITE(6,91)
    WRITE(6,92) HM(K1),HS(K1),SD(K1)
    WRITE(6,92) (A(J,K1),J=1,L)
    WRITE(6,92) (QC(1,I,K1),GC(I,K1),GCL(I,K1),VISC(I,K1),
   1I=1,N)
    WRITE(6,90)
    WRITE(6,93) K
    WRITE(6,92) HM(K),HS(K),SD(K)
    WRITE(6,92) (A(J,K),J=1,L)
    WRITE(6,92) (QC(1,I,K),GC(I,K),GCL(I,K),VISC(I,K),
   1I=1,N)
    GO TO 888
999 STOP
 10 FORMAT(4I4)
 20 FORMAT(5F8.0)
 30 FORMAT(10F8.0)
 50 FORMAT(1H1,3HMY=,I4,5X,3HMX=,I4,5X,2HN=,I4,5X,2HL=,I4)
 51 FORMAT(1H0,19HRW,AL,V,POUHG,ROUS=/1H ,5E13.5)
 52 FORMAT(1H0,6HDH(I)=/(1H ,5E13.5))
 53 FORMAT(1H0,6HTS(I)=/(1H ,5E13.5))
 60 FORMAT(1H0,33HDP(I),Q(1,I),U(I),TW(1,I),TWL(I)=/(1H ,5E13.5))
 65 FORMAT(1H0,29HGC=(GK*TW-GK*A(2))**A(1)/A(3))
 70 FORMAT(1H0,7HA(J,1)=/(1H ,5E13.5))
 80 FORMAT(1H0,4HILL=,I8)
 90 FORMAT(1H0,2HK=/1H ,21HHM(IK),HS(IK),SD(IK)=/1H ,8HA(J,IK)=/
   11H ,41HQC(1,I,IK),GC(I,IK),GCL(I,IK),VISC(I,IK)=)
 91 FORMAT(1H ,4H   0)
 92 FORMAT(1H ,4E13.5)
 93 FORMAT(1H ,I4)
    END


    SUBROUTINE HISENS(Y,X,A,HAA,HM,HMM,HS,HSS,HST,W,YC,YCS,
   1DD,SD,MY,MYT,MX,MXT,N,NT,L,LT,K,KT,L1,K1,EPS,SUB,ILL)
C   METHOD OF NONLINEAR LEAST SQUARE
    DIMENSION Y(MYT,NT),X(MXT,NT),A(LT,K1),HM(K1),HS(K1),W(MYT,NT),
   1YC(MYT,NT,K1),YCS(MYT,NT,L1),DD(LT,L1),SD(K1),HA(20),AS(20),
   2DDS(20)
    IF(KT.GE.1.OR.MY.GE.1.OR.MX.GE.1.OR.N.GE.L.OR.L.GE.1.OR
   1.L.LE.20.OR.L.LT.L1.OR.KT.LT.K1.OR.EPS.GE.0.0) GO TO 200
    ILL=30000
    GO TO 999
200 K=1
    HM(K1)=HM(1)
    HS(K1)=HS(1)
    SDS=0.0
    DO 10 IA=1,L
 10 A(IA,K1)=A(IA,1)
888 CALL SUB(YC,X,A,MY,MYT,MX,MXT,N,NT,L,LT,K,KT,K1,ILL)
    IF(ILL) 300,400,300
300 IF(SDS) 999,999,700
400 D=0.0
    DO 20 I=1,N
    DO 20 J=1,MY
 20 D=D+((Y(J,I)-YC(J,I,K))*W(J,I))**2
    SD(K)=SQRT(D/(N*MY))
    IF(SDS) 999,500,600
500 SDS=SD(K)
```

```
      SD(K1)=SD(K)
      DO 30 I=1,N
      DO 30 J=1,MY
      YCS(J,I,L1)=YC(J,I,K)
   30 YC(J,I,K1)=YC(J,I,K)
      DO 21 IA=1,L
   21 A(IA,K1)=A(IA,1)
      GO TO 777
  600 IF(SDS-SD(K)) 700,800,800
  700 HS(K)=HS(K)*HSS
      IF(HST-HS(K)) 666,999,999
  800 IF(KT-K) 999,999,900
  900 K=K+1
      HM(K)=HM(K-1)*HMM
      HS(K)=HS(1)
      SDS=SD(K-1)
      DO 40 IA=1,L
   40 A(IA,K)=A(IA,K-1)
      DO 50 I=1,N
      DO 50 J=1,MY
   50 YCS(J,I,L1)=YC(J,I,K-1)
  777 DO 60 IA=1,L
      AST=A(IA,K)
      HA(IA)=A(IA,K)*HAA
      A(IA,K)=A(IA,K)+HA(IA)
      CALL SUB(YC,X,A,MY,MYT,MX,MXT,N,NT,L,LT,K,KT,K1,ILL)
      IF(ILL) 999,1000,999
 1000 DO 70 I=1,N
      DO 70 J=1,MY
      YCS(J,I,IA)=YC(J,I,K)
   70 YCS(J,I,IA)=(YCS(J,I,IA)-YCS(J,I,L1))/HA(IA)
   60 A(IA,K)=AST
      LP=L+1
      DO 80 IA1=1,L
      DO 80 IA2=1,L
      DD(IA1,IA2)=0.0
      DO 80 I=1,N
      DO 80 J=1,MY
   80 DD(IA1,IA2)=DD(IA1,IA2)+YCS(J,I,IA1)*YCS(J,I,IA2)
     1*W(J,I)**2
      DO 90 IA=1,L
      DD(IA,IA)=DD(IA,IA)*(1.0+HM(K))
      DD(IA,LP)=0.0
      DO 90 I=1,N
      DO 90 J=1,MY
   90 DD(IA,LP)=DD(IA,LP)+YCS(J,I,IA)*(Y(J,I)-YCS(J,I,L1))
     1*W(J,I)**2
      IF(L-1) 999,333,222
  222 CALL GAUYOS(DD,L,LP,LT,L1,EPS,ILL)
      IF(ILL) 999,1100,999
  333 DD(1,2)=DD(1,2)/DD(1,1)
      ILL=0
 1100 DO 110 IA=1,L
      AS(IA)=A(IA,K)
  110 DDS(IA)=DD(IA,LP)
  666 DO 120 IA=1,L
      DD(IA,LP)=DDS(IA)*HS(K)
  120 A(IA,K)=AS(IA)+DD(IA,LP)
      GO TO 888
  999 RETURN
      END
      SUBROUTINE GAUYOS(A,N,N1,NT,NT1,EPS,ILL)
C     GAUSS-JORDAN METHOD
      DIMENSION A(NT,NT1)
      DO 10 K=1,N
      BIG=ABS(A(K,K))
      IP=K
      K1=K+1
      IF(K1.GT.N) GO TO 14
      DO 11 I=K1,N
      IF(ABS(A(I,K)).LE.BIG) GO TO 11
      BIG=ABS(A(I,K))
      IP=I
   11 CONTINUE
   14 IF(BIG.GE.EPS) GO TO 12
      ILL=1000
      GO TO 999
   12 IF(IP.EQ.K) GO TO 15
      DO 13 J=1,N1
      TEMP=A(K,J)
      A(K,J)=A(IP,J)
   13 A(IP,J)=TEMP
   15 W=A(K,K)
```

```
      DO 20 J=K1,N1
   20 A(K,J)=A(K,J)/W
      DO 30 I=1,N
      IF(I.EQ.K) GO TO 30
      W=A(I,K)
      DO 40 J=K1,N1
   40 A(I,J)=A(I,J)-W*A(K,J)
   30 CONTINUE
   10 CONTINUE
      ILL=0
  999 RETURN
      END


      SUBROUTINE SUB(QC,TW,A,MY,MYT,MX,MXT,N,NT,L,LT,K,KT,K1,ILL)
C     CALCULATION OF NONNEWTONIAN FLOW RATE EQUATION
C     GC=(GK*TW-GK*A(2))**A(1)/A(3)
      DIMENSION QC(MYT,NT,K1),TW(MXT,NT),A(LT,K1)
      COMMON RW,GK
      DO 100 I=1,N
      DT=TW(1,I)-A(2,K)
      IF(DT) 200,300,300
  200 DT=1.0E-5
  300 DTA=DT**(A(1,K)+1.0)
      DTA=2.0*3.1416*GK**A(1,K)*RW**3*DTA/(A(3,K)*TW(1,I)**3)
      DTB=DT**2/(2.0*(A(1,K)+3.0))+A(2,K)*DT/(A(1,K)+2.0)
     1+A(2,K)**2/(2.0*(A(1,K)+1.0))
  100 QC(1,I,K)=DTA*DTB
      ILL=0
      RETURN
      END
```

```
MY=   1      MX=   1      N=   5      L=   3

RW,AL,V,ROUHG,ROUS=
  0.14300E 00  0.26410E 02  0.20000E 02  0.13554E 02  0.10129E 01

DH(I)=
  0.36000E 01  0.68000E 01  0.15000E 02  0.22700E 02  0.34300E 02

TS(I)=
  0.12025E 03  0.49130E 02  0.16050E 02  0.92000E 01  0.53000E 01

DP(I),Q(1,I),U(I),TW(1,I),TWL(I)=
  0.48781E 02  0.16632E 00  0.25889E 01  0.13206E 00 -0.87922E 00
  0.92084E 02  0.40708E 00  0.63367E 01  0.24930E 00 -0.60328E 00
  0.20253E 03  0.12461E 01  0.19397E 02  0.54832E 00 -0.26097E 00
  0.30531E 03  0.21739E 01  0.33839E 02  0.82656E 00 -0.82724E-01
  0.45777E 03  0.37736E 01  0.58740E 02  0.12393E 01  0.93184E-01

GC=(GK*TW-GK*A(2))**A(1)/A(3)

A(J,1)=
  0.10000E 01  0.13206E-01  0.86806E 00

K=
HM(IK),HS(IK),SD(IK)=
A(J,IK)=
QC(1,I,IK),GC(I,IK),GCL(I,IK),VISC(I,IK)=
   0
  0.10000E 01  0.10000E 01  0.29811E 00
  0.10000E 01  0.13206E-01  0.86806E 00
  0.29677E 00  0.13418E 03  0.21277E 01  0.96451E 00
  0.60074E 00  0.26654E 03  0.24258E 01  0.91662E 00
  0.13760E 01  0.60411E 03  0.27811E 01  0.88949E 00
  0.20975E 01  0.91824E 03  0.29630E 01  0.88216E 00
  0.31677E 01  0.13842E 04  0.31412E 01  0.87741E 00

K=
HM(IK),HS(IK),SD(IK)=
A(J,IK)=
QC(1,I,IK),GC(I,IK),GCL(I,IK),VISC(I,IK)=
  25
  0.59605E-07  0.62500E-01  0.57306E-01
  0.12294E 01  0.44837E-01  0.33022E 01
  0.12800E 00  0.71819E 02  0.18562E 01  0.18021E 01
  0.40851E 00  0.20468E 03  0.23111E 01  0.11936E 01
  0.13010E 01  0.61975E 03  0.27922E 01  0.86704E 00
  0.22619E 01  0.10644E 04  0.30271E 01  0.76099E 00
  0.38383E 01  0.17926E 04  0.32535E 01  0.67752E 00
```

加熱デンプン水溶液の流動方程式における各種粘性パラメータ算出に関する研究

久保田清・細川嘉彦・鈴木寛一・保坂秀明

液状食品に関する各種装置の設計ならびに操作を行なっていくためには，流動方程式を設定し，それに含まれる各種粘性パラメータを算出していくことが必要である。本研究は，加熱処理したデンプン水溶液の流動特性を，毛管形粘度計を作製して30，50および70℃において求め，流動方程式における各種粘性パラメータ算出に関する研究を行なったものである。加熱処理した3および5wt％の小麦，トウモロコシ，ジャガイモおよびサツマイモデンプン水溶液を試料とした。

流動方程式 $\gamma=(1/K)(g_c\tau-g_c\tau_y)^n$ に含まれる粘性パラメータ n，τ_y および K を，非線形最小二乗法を使用して算出する電子計算機プログラムを作成した。

加熱処理した3および5wt％の各種デンプン水溶液について，n，τ_y および K の値を計算した結果，$n=1.2\sim1.4$，$\tau_y\fallingdotseq0$ となり，擬塑性流体として取り扱えることが分った。$n=1.3$，$\tau_y=0$ と固定して求めた K の値は，温度の上昇ならびに濃度の減少で小さく変わり，$K=A\ \exp(E/R_gT)$ として表わすことができた。試料とした各種デンプン水溶液に対して，$E\fallingdotseq3\sim7$ kcal／g－mol となる結果が得られた。

本研究成果は，流動特性が複雑な食品などの流動方程式を実験データよりシミュレーションによって得る場合に有用である。